MINEFIELD



Scott Bell and Simon Dowson

The Game

Like many computer games, 'Minefield' is played upon a board or grid. The object of the game is to cross the minefield from square 0,0 to square 9,9 without stepping on a mine. The mines themselves are invisible to the player and the only clue as to their whereabouts is given just before each move, when you are informed of how many mines there are immediately around you.

To make the game a little more difficult there are a number of rocks scattered randomly about the minefield. Unlike the mines, these rocks can be seen and the player is not permitted to enter these squares and must move around them in order to continue.

Should you take more than 20 moves whilst crossing the minefield you will be told that the batteries in your mine detector have become exhausted and that all subsequent moves must be made blindly, i.e. no more warnings of the mines around you will be given.

When you reach square 9,9 or have been blown up, you can if you wish, see a printout of the board showing your route and the positions of the mines. Your path through the minefield is marked so that if you get into difficulties you can backtrack into some safe squares, but be warned, your number of goes is totalling up all of the time. If you have stepped on a mine, your last position will be marked with a 'B'.

If you decide to play another game, a new minefield is produced with the mines and rocks in different positions.

Getting Down to Basics

The program takes up between 5-6 kilobytes of memory when the array and all variables have been set up. By removing the instructions and leaving the bare minimum, the program can be made to run in 2-3 kilobytes.

Only one array, A(11,11), is used and this stores the board information. The array is dimensioned larger than the minefield to allow for attempted moves off the board. The positions of the rocks and the mines are entirely random except that squares (0,0) and (9,9) are always kept clear.

To make the game easier or more difficult, the number of rocks or mines may be altered by changing the FOR — NEXT loops associated with that particular routine.

The program should run on most machines which support BASIC with a few minor changes. For the CBM 'PET 2001', the randomize statement should be omitted, 'RND' changed to 'RND(1)' and 'LEFT(A£,1)' changed to 'LEFT£(A£,1)' etc.

Care should be taken to ensure correct punctuation in the routine that prints out the board. Other than that the program is fairly straightforward . . . . . .

```
10 PRINT
20 PRINT"                              MINEFIELD"
30 PRINT"                              ---------"
40 PRINT"YOU ARE IN SQUARE (0,0)."
50 PRINT"YOUR OBJECTIVE IS TO CROSS THE MINEFIELD AND"
60 PRINT"REACH SQUARE (9,9) IN ONE PIECE!!!!."
70 PRINT"A NUMBER OF MINES HAVE BEEN PLANTED, BUT"
80 PRINT"YOU WILL NOT BE ABLE TO SEE THEM."
90 PRINT"ALSO THERE ARE A FEW ROCKS SCATTERED ABOUT."
100 PRINT"THESE SQUARES ARE OFF LIMITS."
110 PRINT"BEFORE EACH MOVE A WARNING WILL BE GIVEN"
120 PRINT"OF THE NUMBER OF MINES IMMEDIATLEY AROUND YOU."
130 PRINT"BUT YOU WILL NOT BE GIVEN THE POSITIONS."
140 PRINT"IF YOU EXCEED 20 MOVES THEN YOU WILL NO"
150 PRINT"LONGER BE GIVEN THE WARNING."
160 PRINT"YOU MAY MOVE IN ANY DIRECTION EXCEPT"
170 PRINT"DIAGONALLY."
180 PRINT"COMMANDS ARE . . . 'UP', 'DOWN', 'LEFT' OR 'RIGHT'."
190 PRINT:PRINT"GOOD LUCK. . . . . "
200 RANDOMIZE
210 DIM A(11,11)
220 MAT A=ZER:Z=0
230 CLEAR
240 REM***SET UP ROCKS***
250 FOR I=1 TO 10
260 K=INT(9*RND+1):J=INT(9*RND+1)
270 IF A(K,J)=6 THEN GOTO 260
280 IF A(1,1)=6 THEN A(1,1)=0:GOTO 260
290 IF A(10,10)=6 THEN A(10,10)=0:GOTO 260
300 A(K,J)=6:NEXT I
310 REM***SET UP MINES***
320 G=0:N=0
330 FOR I=1 TO 15
340 K=INT(9*RND+1):J=INT(9*RND+1)
350 IF A(K,J)=1 THEN GOTO 340
360 IF A(1,1)=1 THEN A(1,1)=0:GOTO 340
370 IF A(10,10)=1 THEN A(10,10)=0:GOTO 340
380 IF A(K,J)=6 THEN GOTO 340
390 A(K,J)=1:NEXT I:J=1:K=1:A(K,J)=2:A(10,10)=5
400 REM***START OF GAME***
410 GOTO 670
420 PRINT"MOVE NO."G+1;:INPUT"      WHICH DIRECTION";A$:T=0
430 IF LEFT(A$,1)="U"THEN J=J+1:T=3:GOTO 480
440 IF LEFT(A$,1)= "D" THEN J=J-1:T=4:GOTO 480
450 IF LEFT(A$,1)="L"THEN K=K-1:T=5:GOTO 500
460 IF LEFT(A$,1)="R" THEN K=K+1:T=6:GOTO 500
470 PRINT"COMMANDS ARE . . . UP, DOWN, LEFT OR RIGHT." :GOTO 420
480 IF J<1 THEN J=J+1:GOTO 570
490 IF J>10 THEN J=J-1:GOTO 570
500 IF K<1 THEN K=K+1:GOTO 570
510 IF K>10 THEN K=K-1:GOTO 570
520 IF A(K,J)=6 AND T=3 THEN J=J-1:GOTO 590
530 IF A(K,J)=6 AND T=4 THEN J=J+1:GOTO 590
540 IF A(K,J)=6 AND T=5 THEN K=K+1:GOTO 590
550 IF A(K,J)=6 AND T=6 THEN K=K-1:GOTO 590
560 G=G+1:GOTO 600
570 PRINT"STAY ON THE MINEFIELD, COWARD!!!"
580 GOTO 420
590 PRINT"YOU CAN'T MOVE INTO A SQUARE WHERE THERE ARE ROCKS.":GOTO 420
600 IF T=3 THEN A(K,J-1)=4
610 IF T=4 THEN A(K,J+1)=4
620 IF T=5 THEN A(K+1,J)=4
630 IF T=6 THEN A(K-1,J)=4
640 IF K=10 AND J=10 THEN 1000
650 IF A(K,J)=1 THEN A(K,J)=7:PRINT"BOOM!!!!!. . . . YOU'RE DEAD":GO
660 A(K,J)=2
670 REM***DETECTS IF ANY MINES AROUND YOU***
680 V=0
690 FOR A=K-1 TO K+1
700 FOR B=J-1 TO J+1
710 IF A(A,B)=1 THEN V=V+1
720 NEXT B
730 NEXT A
740 REM***PRINT BOARD***
750 FOR A=10 TO 1 STEP -1
760 PRINT"  ---------------------------------------------"
770 PRINT A-1;
780 FOR B=1 TO 10
790 PRINT"!";
800 IF A(B,A)=0 THEN A$="   "
810 IF A(B,A)=1 THEN A$="   "
820 IF A(B,A)=1 AND Z=1 THEN A$=" M "
830 IF A(B,A)=4 THEN A$=" * "
840 IF A(B,A)=7 THEN A$=" B ":N=1
850 IF A(B,A)=2 THEN A$=" Y "
860 IF A(B,A)=5 THEN A$=" H "
870 IF A(B,A)=6 THEN A$=" R "
880 PRINT A$;:NEXT B:PRINT"!":NEXT A
890 PRINT"  ---------------------------------------------"
900 PRINT"     0   1   2   3   4   5   6   7   8   9   "
910 IF A(K,J)=5 THEN GOTO 1030
920 IF G<>20 THEN GOTO 970
930 PRINT"THE BATTERIES IN YOUR MINE-DETECTOR"
940 PRINT"ARE EXHAUSTED.          "
950 PRINT"YOU WILL HAVE TO NAVIGATE BLINDLY"
960 PRINT"FROM NOW ON   HA HA HA"
970 IF N=1 THEN GOTO 1080
980 IF G<20 THEN PRINT:PRINT"THERE ARE"V"MINE(S) AROUND YOU."
990 GOTO 420
1000 PRINT"WELL DONE. . . . . WOULD YOU LIKE TO SEE"
1010 INPUT"YOUR ROUTE AND WHERE THE MINES WERE";A$
1020 IF LEFT(A$,1)="Y" THEN Z=1:GOTO 740
1030 PRINT"IT TOOK YOU"G"MOVES AS WELL."
1040 INPUT"ANOTHER GAME";B$
1050 IF LEFT(B$,1)="Y"THEN GOTO 210
1060 PRINT"O.K. THEN. . . . . GOOD-BYE":GOTO 1100
1070 PRINT"HARD LINES, WOULD YOU LIKE TO SEE":GOTO 1010
1080 PRINT"THE LETTER 'B' IS WHERE YOU STOOD"
1090 PRINT"ON A MINE.":GOTO 1030
1100 END
```